
Einleitung

Im ersten Teil habe ich beschrieben, welche Möglichkeiten Access für ein einfaches Errorhandling bietet. Diese Mechanismen reichen in der Regel aus, um die Stabilität und Zuverlässigkeit der Anwendung soweit zu sichern, dass ein relativ ungestörter Betrieb der Anwendung möglich ist.

In der Praxis verlangen die Kunden oft, dass zusätzlich zu dem bereits beschriebenen Vorgehen erweiterte Möglichkeiten zur Behandlung und Untersuchung von aufgetretenen Fehlern ergriffen werden. Dafür sind komplexere Fehlerbehandlungen notwendig.
In diesem Artikel werde ich auf folgende Punkte eingehen:

· Reagieren auf bestimmte Fehlernummern im Errorhandling

· Fehler-Protokollierung (Schreiben der Fehlerinformation in eine Datei oder in die Datenbank) und Durchleuchten des Err.-Objektes und der Errors-Collection.

· Informationen darüber, in welchem Formular und in welcher Prozedur der Fehler aufgetreten ist

· Information über die Zeile, in der der Fehler aufgetreten ist

Reagieren auf bestimmte Fehlernummern im Errorhandling

Eine häufige Anforderung an das Errorhandling ist, dass abhängig vom aufgetretenen Fehler unterschiedliche Aktionen ausgeführt werden sollen. Es müssen nicht alle Fehler, die auftreten, für den Benutzer als Fehler sichtbar werden. Oftmals ist es möglich, den aufgetretenen Fehler per Code zu erkennen und entsprechend unterschiedliche Reaktionen auszuführen.

Rechnen in Zahlenfeldern

Ein einfaches Beispiel soll dies veranschaulichen: Ein Formularfeld sei an ein numerisches Feld in der Datenbank gebunden. In dieses Feld können per Definition demzufolge nur Zahlenwerte eingegeben werden, alle anderen Eingaben werden von Access mit einer internen Fehlermeldung quittiert.

Der Benutzer hat aber oftmals das Bedürfnis, dass er statt Zahlenwerten mathematische Ausdrücke eingeben möchte, z.B. möchte er einen bestimmten Prozentanteil eines Wertes errechnen oder zwei andere Werte zusammenzählen. Der Benutzer ist dann überrascht, dass er zwar einen Computer hat, mit diesem aber nur auf Umwegen über den Rechner (calc.exe) oder ein anderes Programm wie MS Excel® rechnen kann, nicht jedoch direkt in einem Zahlenfeld eine Formel eingeben darf.

Die Eingabe von "=156*0.025" (er möchte 2.5% von 156 errechnen), quittiert die Anwendung mit einem Fehler: "Sie haben einen Wert eingegeben, der für dieses Feld nicht zulässig ist.".

Wenn wir diesen Fehler im Form_Error Ereignis untersuchen, sehen wir, dass die Fehlernummer 0 ist, der Fehler aber über die Variable DataErr zurückgegeben wird. Hier haben wir also die Möglichkeit auf unterschiedliche Fehlernummern verschieden zu reagieren, genau so, wie wir das auch in einer Error Sektion einer Prozedur machen können.

Mit folgenden wenigen Codezeilen können wir Berechnungen zulassen:

Private Sub Form_Error(DataErr As Integer, Response As Integer)

  Dim ctl As Control

  Dim strInput As String

  Select Case DataErr

    Case 2113

      Set ctl = Screen.ActiveControl

      On Error Resume Next

      If Left(ctl.Text, 1) = "=" Then

        strInput = ctl.Text

        ctl.Text = Eval(Mid(strInput, 2))

        If ctl.Text <> strInput Then

          Response = acDataErrContinue

        Else

          Response = acDataErrDisplay

        End If

      Else

        Response = acDataErrDisplay

      End If

    Case Else

      Response = acDataErrDisplay

  End Select

End Sub

Im Falle, dass DataErr = 2113 ist, setzen wir eine Referenz auf das aktive Control und kontrollieren, ob der Benutzer die Eingabe in dieses Feld mit "=" begonnen hat. Falls ja, gehen wir davon aus, dass er dieses Feld errechnen lassen will. Wir rufen also mit der gemachten Eingabe (ohne "=") die Eval() Funktion auf und schreiben das Ergebnis in das Text-Property des aktiven Controls zurück (Anmerkung: Access wird in diesem Fall immer einen Fehler zurückmelden, jedoch den Wert richtig zurückschreiben). Falls sich der Wert des Text-Properties verändert hat, konnte die Eval() Funktion den Text interpretieren und das Ergebnis wurde ins Feld geschrieben. Wir weisen in diesem Fall Access an, die eingebaute Fehlermeldung zu unterdrücken. In allen anderen Fällen soll sich Access so verhalten, wie bisher.

Wenn wir diese einfache Formular-Fehlerbehandlungsroutine in irgendeinem Formular einbauen, er​mög​lichen wir dem Benutzer in Formularfeldern zu rechnen. Hier eine Übersicht, was gerechnet wird.

	Feldtyp
	Eingabe
	Resultat

	Text
	=156*0.025
	"=156*0.025"

	Double, numerisch
	=156*0.025
	3.9

	
	=2.5% von 156
	Eingebaute Fehlermeldung

	Datum, Zeit
	=Date() + 7
	#[Heute in einer Woche]#

	
	30.2.2003
	Eingebaute Fehlermeldung


Anmerkungen

· Die Eval() Funktion kennt nur englische Ausdrücke. Die Funktion Jetzt() oder Datum() sind also nicht bekannt und müssen durch die Englischen Ausdrücke (Now() oder Date()) ersetzt werden.

· Eval() ignoriert die Ländereinstellungen für Zahlenformate. Falls das Komma "," als Dezimaltrennzeichen verwendet werden soll, muss dieses vor dem Aufruf von Eval() durch einen Dezimalpunkt "." ersetzt werden. Ebenfalls müssen allfällig eingegebene Tausender Trennzeichen entfernt werden, wenn diese nicht mit der englischen Notation kompatibel sind.

Fehler Protokollierung

Eine der wichtigsten Schritte zur Verbesserung der Stabilität von Anwendungen ist die Protokollierung aller aufgetretenen Fehler, um diese zu einem späteren Zeitpunkt zu untersuchen, Auswertungen über die Fehlerursachen erstellen zu können und Korrekturen an der Anwendung vorzunehmen, die diese Fehler unterbinden.

Der einfache Errorhandler aus dem ersten Teil

On Error GoTo ProcErr

…

ProcExit:

  Exit Sub 'oder Function

…

ProcErr:

  MsgBox "Err.Number" & ": " & Err.Description

  Resume ProcExit

End Function 'oder Sub

rsp. die letzte Sektion ab dem Label ProcErr: soll hier als Basis für die Fehler Protokollierung dienen.

Statt die Fehlermeldung direkt als Meldung auszugeben, können wir hier eine Fehlerbehandlungs​routine aufrufen.

…

ProcErr:

  Call ErrorHandler()

  Resume ProcExit

End Function 'oder Sub

Diese Funktion selber kann nun die Fehlerinformation zusätzlich zum Ausgeben der Meldung für den Benutzer in ein Protokoll schreiben. Damit auch Fehler weggeschrieben werden können, wenn es Probleme beim Zugriff auf die Tabellen gibt (z.B. Server mit der Backend ist nicht verfügbar), empfiehlt es sich, diese Information in ein lokales Textfile zu schreiben. Es ist unter Umständen sinnvoll, dass zuerst versucht wird, in eine zentrale ErrorLog Tabelle auf dem Backend zu schreiben und nur wenn das Backend nicht verfügbar ist, die lokale Text-Datei anzulegen, damit die Fehler im Normalfall zentral gesammelt und ausgewertet werden können.

Sichern der Fehlerinformationen

Um die Fehlerinformation nicht zu verlieren, muss der ErrorHandler() eine spezielle Fehlerbehandlung aufweisen. Bevor die eigentliche Fehlerbehandlung aktiviert wird, müssen die Fehlerinformationen gesichert werden. Die Prozedur darf also auf keinen Fall mit einem On Error … Statement beginnen, da dieses Statement die Fehlerinformation vernichten und den Fehler zurücksetzen würde, womit eine Protokollierung der aufgetretenen Fehler nicht mehr möglich ist.

Die Fehlerinformation ist in verschiedenen Eigenschaften des Err-Objektes abgelegt. Um diese zwischenzuspeichern kann eine Typen-Deklaration verwendet werden. Die Error-Objekte der Errors-Collection weisen eine ähnliche Struktur auf. Zusätzlich verfügt es über die Properties HelpContext und HelpFile, kennt aber das LastDLLError-Property nicht.

Vielfach ist es ausreichend, eine reduzierte Version eines Datentypen für die Sicherung der Fehlerinformation zu definieren, welche die gemeinsamen Felder beinhaltet, da die weggelassenen Felder selten zur Fehlerbehebung beitragen.

Type ErrInfo

  Nbr As Long    'Err.Number

  Dsc As String  'Err.Description

  Src As String  'Err.Source

End Type

Mit dieser Typendeklaration kann nun zu Beginn der Funktion die noch aktuelle Fehlerinformation gesichert werden. Dabei kann der Typ verwendet werden, um einen Array zu deklarieren, der dann anhand der Anzahl der Fehler in der Errors-Collection redimensioniert werden kann. So haben wir die Möglichkeit, das Err-Objekt in das 0-te Array Element und die Error Objekte in die folgenden Array-Variablen abzufüllen.

Public Function ErrorHandler()

  Dim errSave() As ErrInfo

  Dim I As Long

  ReDim errSave(0)

  With errSave(0)

    .Nbr = Err.Number

    .Dsc = Err.Description

    .Src = Err.Source

  End With

  If Errors(Errors.Count - 1).Number = Err.Number Then

    ReDim Preserve errSave(0 To Errors.Count - 1)

    For I = Errors.Count - 1 To 0 Step -1

      With errSave(I)

        .Nbr = Errors(I).Number

        .Dsc = Errors(I).Description

        .Src = Errors(I).Source

      End With

    Next I

  End If

  …

End Function

Bei der Sicherung der Informationen sind einige Tücken zu umschiffen. Die Errors Collection wird nicht zurückgesetzt, wenn ein neuer Fehler auftritt, der diese nicht benutzt. Der letzte in der Errors Collection enthaltene Fehler (Errors.Count - 1) ist der zurückgemeldete Fehler, der im Err-Objekt enthalten ist. Falls die Err.Number nicht mit der Nummer des letzten Fehlers in der Errors-Collection identisch ist, muss davon ausgegangen werden, dass diese nicht benutzt wurde. Sie enthält daher noch alte Error-Objekte eines früheren Fehlers, die nicht gelöscht wurden. Sie darf in diesem Fall nicht ausgewertet werden. Falls die Fehlernummer identisch ist, enthält dagegen die Err.Description oftmals keinen aussagekräftigen Text, sondern irgendetwas wie "ODBC Aufruf fehlgeschlagen". Der aussagekräftigere Text für diese Fehlernummer befindet sich in diesem Fall im letzten Objekt der Errors-Collection. Daher führt nachfolgender Ansatz zu vernünftigen Fehlerinformationen:

Zuerst wird das Err-Objekt gesichert. Anschliessend wird kontrolliert, ob das letzte Error-Objekt der Errors-Collection mit dem Err-Objekt identisch ist (Vergleich über .Number). Falls das der Fall ist, wird die Information des Err-Objekts mit der Information der letzten Error-Objektes überschrieben. Alle weiteren, in der Errors-Collection enthaltenen Fehler-Objekte werden ebenfalls gesichert, da diese unter Umständen nützliche weitere Informationen zum Fehler enthalten.

Nachdem wir die relevanten Fehlerinformationen in Variablen gesichert haben, können wir die Fehlerbehandlung auch im ErrorHandler() starten, den wir z.B. für Fehler beim Schreiben des Fehler-Protokolls benötigen.

Zusätzlich benötigte Informationen für ErrorHandler()

Die Informationen, die wir bis jetzt gesammelt haben, sind für die nachträgliche Untersuchung des Fehlers selten ausreichend. Für eine detaillierte Fehleruntersuchung benötigen wir zusätzliche Information wie:

· Woher wurde der ErrorHandler() aufgerufen

· Name der aufrufenden Prozedur

· Name des Moduls, in dem diese Prozedur enthalten ist

· Die Zeile, in der der Fehler aufgetreten ist

· Die Version der Anwendung

· Der Benutzer, der angemeldet war

· Der Zeitpunkt, wann der Fehler aufgetreten ist

Leider bietet Access bisher keine eingebaute Möglichkeit, direkt den Namen der Prozedur und des Moduls auszulesen. Wir müssen diese Information also beim Aufruf von ErrorHandler() übergeben. Hingegen kann Access auf Umwegen die Zeilennummer ausfindig machen, in der der Fehler aufgetreten ist. Wie das geht, wird später beschrieben. Wichtig in diesem Zusammenhang ist, dass diese Information bei einem On Error – Statement ebenfalls verloren geht und daher vorgängig ebenfalls gesichert werden muss.

Standardisierter ErrorHandler()

Mit diesen Anforderungen bauen wir den ErrorHandler() folgendermassen aus:
Public Function ErrorHandler(ModuleName As String, _

                             ProcName As String)

  Dim errSave() As ErrInfo

  Dim I As Long

  Dim strUserName As String

  Dim dteError As Date

  Dim strErrMsg As String

  Dim varErrorLine As Variant

  Dim strVersion As String

  'save current error information

  ReDim errSave(0)

  With errSave(0)

    .Nbr = Err.Number

    .Dsc = Err.Description

    .Src = Err.Source

  End With

  If Errors(Errors.Count - 1).Number = Err.Number Then

    ReDim Preserve errSave(0 To Errors.Count - 1)

    For I = Errors.Count - 1 To 0 Step -1

      With errSave(I)

        .Nbr = Errors(I).Number

        .Dsc = Errors(I).Description

        .Src = Errors(I).Source

      End With

    Next I

  End If

  varErrorLine = Erl()

  'initialize error handler

  On Error GoTo PROCERR

  'get additional information

  strVersion = getVersion()

  strUserName = CurrentUser()

  dteError = Now()

  'create a error message to display

  strErrMsg = "Fehler in " & ModuleName & "." & ProcName & _

              " in Zeile Nummer " & varErrorLine & vbCrLf & _

              "bei " & strUserName & " um " & dteError & vbCrLf & _

              errSave(0).Nbr & ": " & errSave(0).Dsc

  'inform the user about the error

  MsgBox strErrMsg, vbExclamation

  'log the error

  Call logError(errSave, _

                ModuleName, _

                ProcName, _

                varErrorLine, _

                strUserName, _

                dteError)

PROCEXIT:

  Exit Function

PROCERR:

  'let the user know something went wrong here

  MsgBox "Error in ErrorHandler() - " & _

         Err.Number & ": " & Err.Description

  Resume PROCEXIT

End Function

Dieser Errorhandler gibt die erweiterten Fehlerinformationen als Meldung aus und ruft die Protokol​lierungs​routine mit der zwischengespeicherten Fehlerinformation auf. Die Protokollierungs​funktion ihrerseits muss keine eigene Fehlerbehandlung beinhalten. Hier auftretende Fehler werden an ErrorHandler() zurückgemeldet und von diesem behandelt.

Funktion getVersion()

Zur Fehlerinformation gehören, wie bereits angedeutet, ebenfalls die Versionsinformation und die Zeilennummer, in der der Fehler aufgetreten ist. Um die Version einzulesen, empfiehlt es sich, eine zentrale Funktion getVersion() zu benutzen, welche vereinfacht folgendermassen aussehen könnte:

Public Function getVersion()

  getVersion = "0.0.1"

End Function

Selbstverständlich kann hier eine beliebige andere Funktion verwendet werden, die z.B. die Version aus der Datenbank ausliest oder den Wert einer global definierten Konstante zurück liefert. Es ist dabei zu beachten, dass die Funktion getVersion() idealerweise keine Datenbank-Zugriffe machen sollte, um das Risiko zu vermindern, dass in dieser Funktion selber Fehler auftreten. Daher ist die Verwendung von Konstanten oder die hart kodierte Version von getVersion() für die Fehlerbehandlung zu bevorzugen.

Die Erl()-Funktion liefert die Zeilennummer in der der Fehler aufgetreten ist

Zur Ermittlung der Zeile, in der der Fehler aufgetreten ist, vorab nur soviel, dass VBA dafür eine Funktion Namens Erl() zur Verfügung stellt. Die Version und die Zeilennummer werden im ErrorHandler() selbst ermittelt, müssen also nicht übergeben werden.

Schreiben des Fehlers in eine Datei mittels logError()

Nun benötigen wir die Protokollierungsfunktion logError(). Diese soll für unser Beispiel alle vorhandene Fehlerinformation in ein Textfile am Ort des Frontends wegschreiben und nicht versuchen, die Daten zuerst auf eine zentrale Tabelle im Backend abzulegen. Die Datei legen wir als CSV-Datei aus, damit diese später einfach für Auswertungen mittels Einbinden in Access weiterverwendet werden kann. Falls die Datei noch nicht existiert, soll diese angelegt und in der ersten Zeile die Feldnamen festgelegt werden. Die Fehlerinformationen werden an die Datei angefügt.

Private Function logError(arrError() As ErrInfo, _

                          ModuleName As String, _

                          ProcName As String, _

                          ErrorLine As Variant, _

                          strUserName As String, _

                          dteError As Date)

  Dim strLogFile As String

  Dim lngFile As Long

  Dim I As Long

  strLogFile = Left(CurrentDb.Name, _

               Len(CurrentDb.Name) - _

               Len(Dir(CurrentDb.Name))) _

               & "ErrorLog.csv"

  lngFile = FreeFile

  'create logfile if it doesn't exist yet

  If Dir(strLogFile) = "" Then

    Open strLogFile For Append As lngFile

    Print #lngFile, """DateTime"";" & _

                    """Module"";" & _

                    """Procedure"";" & _

                    """Line"";" & _

                    """UserName"";" & _

                    """ErrorNumber"";" & _

                    """ErrorDescription"";" & _

                    """ErrorSource"""

    Close lngFile

  End If

  'open logfile and append all errors

  Open strLogFile For Append As lngFile

  For I = LBound(arrError()) To UBound(arrError())

    With arrError(I)

      Print #lngFile, _

            """" & dteError & """" & _

            ";""" & ModuleName & """" & _

            ";""" & ProcName & """" & _

            ";""" & ErrorLine & """" & _

            ";""" & strUserName & """" & _

            ";""" & .Nbr & """" & _

            ";""" & .Dsc & """" & _

            ";""" & .Src & """"

    End With

  Next I

  'close the logfile

  Close lngFile

End Function

Aufruf von ErrorHandler()

Der Aufruf der Funktion ErrorHandler() erfolgt aus den Fehlerbehandlungssektionen der einzelnen Prozeduren. Für den Aufruf des ErrorHandler() müssen wir die Parameter Modul- und Prozedurname übergeben, da diese von VBA nicht direkt ermittelt werden können.

Der ErrorHandler() kann nicht als Form_Error-Property oder Report_Error-Property aufgerufen werden, da in diesem Fall das Err-Objekt und die Errors-Collection nicht existieren, rsp. nicht an die Funktion weitergegeben werden. Für Formular und Bericht OnError Prozeduren sollten spezielle Prozeduren eingesetzt werden, wie dies zu Beginn des Artikels beschrieben wurde.

In den Prozeduren erfolgt der Aufruf folgendermassen:

Private Sub Form_BeforeUpdate(Cancel As Integer)

  …

PROCERR:

  Call ErrorHandler(Me.Name, "Form_BeforeUpdate")

  Resume PROCEXIT

End Sub

Hier ist zu beachten, dass Me.Name nur in gebundenen Klassenmodulen von Formularen und Berichten funktioniert. Falls die Prozedur sich in einem ungebundenen Standard Modul befindet, muss stattdessen:

Call ErrorHandler("ModulName", "Form_BeforeUpdate")

übergeben werden. Der Modul- und Prozedurname müssen in diesem Fall also als String übergeben werden.

In welcher Prozedur ist der Fehler aufgetreten

Wie bereits erwähnt, bietet Access keine Möglichkeit, den Modul und Prozedurnamen direkt auszulesen, in denen der Fehler aufgetreten ist. Auf Umwegen über die Zeile, in der der Fehler aufgetreten ist, wäre dies machbar, bedingt jedoch erheblichen Aufwand mit der Zeilennummerierung und für die Vorbereitung der Anwendung vor der Verteilung. Alle Zeilennummern müssten über die gesamte Anwendung eindeutig sein und eine Cross-Referenz zwischen Zeilennummern, Prozeduren und Modulen muss erstellt werden, damit von der Zeilennummer auf die Prozedur und das Modul geschlossen werden kann. Es ist daher einfacher, den Modul- rsp. Prozedurname direkt dem ErrorHandler() als Parameter zu übergeben. Alternativen bieten erweiterte Tools, wie z.B. die Total Access CodeTools der FMS Inc. (http://www.fmsinc.com)

In welcher Zeile ist der Fehler aufgetreten

VBA beinhaltet eine Funktion, mit der man herausfinden kann, in welcher Zeile der Fehler aufgetreten ist. Diese Erl() Funktion ist wenig bekannt und sehr schlecht dokumentiert. Eine Grundvoraussetzung für diese Funktion ist, dass alle VBA Code Zeilen nummeriert sind. Zeilennummern sind in VBA nichts anderes als Label. Die Erl() Funktion macht nichts anderes als im VBA Interpreter das letzte numerische Label abzufragen, das eingelesen wurde und zurück zu liefern.

Seit Access 97 ist es glücklicherweise nicht mehr notwendig, dass die Label einer Anwendung über das gesamte Modul hinweg eindeutig sind. Die Label müssen nur noch innerhalb einer Prozedur (Function oder Sub) eindeutig sein. Damit ergibt sich die Möglichkeit, die Zeilen innerhalb einer Prozedur jeweils vom gleichen Startwert hoch zu zählen, ohne schnell überschneidende Bereiche zu erhalten. Da es nicht einfach ist, die Nummern im Code händisch zu verwalten, empfiehlt sich für die Nummernvergabe im VBA Code die Verwendung eines Werkzeuges (z.B. MDEC-97 von www.psp-online.com oder die bereits erwähnten Code Tools der FMS Inc. www.fmsinc.com) vor der Aus​lieferung der Anwendung, welche die Zeilennummern vor der MDE Erstellung einfügt.

Diese Nummerierung kann auch selber implementiert werden. Als Beispiel habe ich ein Modul in die Beispiel-Datenbank eingefügt, welches ermöglicht, einzelne Module zu nummerieren. Dieses Modul enthält alle Bestandteile, die notwendig sind, um eine Anwendung mit Zeilennummern im VBA Code zu versehen.

Die Untersuchung von Fehlern anhand der Modul- und Prozedurnamen, sowie der Zeilennummern nützt allerdings nur dann etwas, wenn die Version der Anwendung identisch ist. Es ist daher zu empfehlen, dass der gesamte VBA Code vor der Auslieferung nach dem Einfügen der Zeilennummern exportiert wird, damit der Fehler im Code später wieder lokalisiert werden kann. Daher ist es wichtig, dass auch die Version, bei der der Fehler aufgetreten ist, mit der Fehlerinformation protokolliert wird, da sonst keine ausreichende Fehleruntersuchung möglich ist. Auch diese Arbeiten können oben genannte Werkzeuge teilweise selbständig erledigen und damit eine Archivierung des Codes für eine spätere Fehleruntersuchung sicherstellen.

Beispielanwendung EEH.MDB

Die beiliegende Beispielanwendung EEH.MDB ist eine Access 97 Anwendung. Diese beinhaltet:

· Ein Formular (aForm), in welchem in Zahlen und Datumsfeldern gerechnet werden kann

· Ein Formular (FehlerhaftesFormular), mit dem der ErrorHandler(), den wir hier entwickelt haben, ausgetestet werden kann. Dieses Formular enthält 3 Schaltflächen zur Auslösung unterschiedlicher Fehler (VBA Fehler (Division durch 0), ODBC Fehler (Aufruf der Fehlerhaften Query), Access Fehler (Versuchtes Öffnen eines nicht existierenden Formulars))

· Das Modul mdlLineNumbering, mit dem Module (Standard und Klassenmodule) einer Anwendung mit Zeilennummern versehen werden können, welche im Fehlerfall von der Erl()-Funktion zurückgeliefert werden

Der Code der EEH.MDB darf beliebig weiterverwendet werden. Gewissen Restriktionen unterliegt das Modul mdlLineNumbering, welches nur als Ganzes zusammen mit dem Kommentar in Anwendungen eingesetzt, rsp. veröffentlicht werden darf.

